[Defs] Useful function definitions

In[ ]:=
```
Efunc[μ_, Γ_, a_, x_] := ℯ^(-x^2/(2 μ)) Sum[DiracDelta[x - (s + a) Γ], {s, -∞, ∞}];

Efunct[μ_, Γ_, a_, x_] := ℯ^(-x^2/(2 μ)) Sum[ℯ^(2 π ⅈ a s) DiracDelta[x - s Γ], {s, -∞, ∞}];

Gauss[v_, x_] := 1/Sqrt[2 π v] ℯ^(-x^2/(2 v));

Θ[a_, b_, z_, τ_] := Exp[π ⅈ τ a^2 + 2 π ⅈ a (z + b)] EllipticTheta[3, π (z + τ a + b), Exp[π ⅈ τ]];
Θs[a_, b_, z_, τ_, s_] := Exp[π ⅈ τ (s + a)^2 + 2 π ⅈ (s + a) (z + b)];

αd[d_] := Sqrt[2 π/d];

Λ[vq_, vp_] := 1 - 4 vq vp;

norm[vq_, vp_, Γ_, j_, d_] := Re[Θ[j/d, 0, 0, (2 ⅈ Γ^2/π) (vp^2/Λ[vq, vp])] × Θ[0, 0, 0, (2 π ⅈ/Γ^2) (vq^2/Λ[vq, vp])] + Θ[j/d + 1/2, 0, 0, (2 ⅈ Γ^2/π) (vp^2/Λ[vq, vp])] × Θ[0, 1/2, 0, (2 π ⅈ/Γ^2) (vq^2/Λ[vq, vp])]];

EconvG[μ_, Γ_, a_, v_, x_] := Gauss[v, x] × Θ[a, 0, Γ x/(2 π ⅈ v), ⅈ Γ^2 (1 + v/μ)/(2 π v)];

EtconvG[μ_, Γ_, a_, v_, x_] := Gauss[v, x] × Θ[0, a, ⅈ Γ x/(2 π v), ⅈ Γ^2/(2 π v)];

EconvGs[μ_, Γ_, a_, v_, x_, s_] := Gauss[v, x] × Θs[a, 0, Γ x/(2 π ⅈ v), ⅈ Γ^2 (1 + v/μ)/(2 π v), s];

EtconvGs[μ_, Γ_, a_, v_, x_, s_] := Gauss[v, x] × Θs[0, a, ⅈ Γ x/(2 π v), ⅈ Γ^2/(2 π v), s];
(*Note : σ^2 = 0.5 means 0 squeezing so 0 < σ^2 (variance) < 0.5 *)

ProductTerm[q_, p_, d_, i_, j_, vq_, vp_, Γ_, parity_] := Chop[EconvG[Λ[vq, vp]/(4 vp), Γ, (i + j)/(2 d) + parity/2, vq, q]] × Chop[EtconvG[Λ[vq, vp]/(4 vq), π Λ[vq, vp]/Γ, (i - j)/(2 d) + parity/2, vp, p]];

Wigner[q_, p_, d_, i_, j_, vq_, vp_, Γ_] :=
  1/Sqrt[norm[vq, vp, Γ, i, d] × norm[vq, vp, Γ, j, d]] Chop[ProductTerm[q, p, d, i, j, vq, vp, Γ, 0] + ProductTerm[q, p, d, i, j, vq, vp, Γ, 1]];
(* Wigner for vacuum *)
Vacner[q_, p_] := Gauss[1/2, q] × Gauss[1/2, p];
(* Wigner for Fock states *)
Fockner[q_, p_, n_] := (-1)^n Gauss[1/2, q] × Gauss[1/2, p] LaguerreL[n, 2 (q^2 + p^2)];

(* After beamsplitter given Bololiubov B = {{t, ⅈ r},{ⅈ r, t}} with t = √η *)
(* The quadratures will mix according to {q1,p1,q2,p2} as {{t,0,0,-r},{0,t,r,0},{0,-r,t,0},{r,0,0,t}} *)

B[t_, r_] := {{t, 0, 0, -r}, {0, t, r, 0}, {0, -r, t, 0}, {r, 0, 0, t}};
{Q1, P1, Q2, P2} = B[t, r].{q1, p1, q2, p2};
(* Gaussian Integral *)

GaussianIntegral[M_] := (2 ℯ^((M[[1,3]] M[[2,1]]^2 - M[[1,2]] M[[2,1]] M[[2,2]] + M[[1,1]] M[[2,2]]^2 + M[[1,2]]^2 M[[3,1]] - 4 M[[1,1]] M[[1,3]] M[[3,1]])/(M[[2,2]]^2 - 4 M[[1,3]] M[[3,1]])) π)/(Sqrt[-4 M[[1, 3]] + M[[2,2]]^2/M[[3,1]]] Sqrt[-M[[3, 1]]]);

ExpCoefficientsToMatrix[expr_, x_, p_] := CoefficientList[Exponent[expr, ℯ], {x, p}, {3, 3}];

(* Trunated Gaussian integrals
  ∫_{-Δ/2}^{Δ/2} ℯ^(c x^2 + b x + a) ⅆx *)

TruncatedGaussInt[v_, Δ_] := (ℯ^(v[[1]] - v[[2]]^2/(4 v[[3]])) Sqrt[π] (-Erfi[(v[[2]] - Δ v[[3]])/(2 Sqrt[v[[3]]])] + Erfi[(v[[2]] + Δ v[[3]])/(2 Sqrt[v[[3]]])]))/(2 Sqrt[v[[3]]]);
```

[Calc] W_{gkp}(q1, p1) W_0(q2, p2) → (EcovG[Q1]Gauss[v21/2, P2])(EtconvG[P1] Gauss[v2=1/2, Q2])

```
(* E convolution G *)
```

In[ ]:=
```
EconvGs[μ, Γ, a, v1, Q1, s] × Gauss[v2, P2]
```

Out[ ]=
$$\frac{e^{-\frac{(-p2\,r+q1\,t)^2}{2\,v1}-\frac{(q1\,r+p2\,t)^2}{2\,v2}+\frac{(a+s)\,(-p2\,r+q1\,t)\,\Gamma}{v1}-\frac{(a+s)^2\,\Gamma^2\left(1+\frac{v1}{\mu}\right)}{2\,v1}}}{2\,\pi\,\sqrt{v1}\,\sqrt{v2}}$$

```
(* Etilde convolution G *)
```

In[ ]:=
```
EtconvGs[μ, Γ, a, v1, P1, s] × Gauss[v2, Q2]
```

Out[ ]=
$$\frac{e^{-\frac{(q2\,r+p1\,t)^2}{2\,v1}-\frac{(-p1\,r+q2\,t)^2}{2\,v2}-\frac{s^2\,\Gamma^2}{2\,v1}+2\,i\,\pi\,s\left(a+\frac{i\,(q2\,r+p1\,t)\,\Gamma}{2\,\pi\,v1}\right)}}{2\,\pi\,\sqrt{v1}\,\sqrt{v2}}$$

[Calc] Tracing out mode 2 - vacuum loss

```
(* For EconvG term - after beamsplitting it is a function of purely q1 and p2. Tracing out mode 2 means integrating over p2 alone. Similar
   reasoning for Etilde conv G and integrating q2. *)
ClearAll[a, b, c];

{cg, bg, ag} = CoefficientList[-(-p2 r + q1 t)^2/(2 v1) - (q1 r + p2 t)^2/(2 v2) + (a + s) (-p2 r + q1 t) Γ/v1 - (a + s)^2 Γ^2 (1 + v1/μ)/(2 v1), {p2}];

Sqrt[π]/(2 π Sqrt[v1] Sqrt[v2]) ℯ^(bg^2/(4 (-ag)) + cg)/Sqrt[-ag] (* Gaussian integral over momentum p2 *)
```

Out[ ]=
$$\frac{e^{-\frac{q1^2\,t^2}{2\,v1}-\frac{q1^2\,r^2}{2\,v2}+\frac{q1\,(a+s)\,t\,\Gamma}{v1}+\frac{\left(\frac{q1\,r\,t}{v1}-\frac{q1\,r\,t}{v2}-\frac{r\,(a+s)\,\Gamma}{v1}\right)^2}{4\left(\frac{r^2}{2\,v1}+\frac{t^2}{2\,v2}\right)}-\frac{(a+s)^2\,\Gamma^2\left(1+\frac{v1}{\mu}\right)}{2\,v1}}}{2\,\sqrt{\pi}\,\sqrt{v1}\,\sqrt{\frac{r^2}{2\,v1}+\frac{t^2}{2\,v2}}\,\sqrt{v2}}$$

In[ ]:=
```
Simplify[CoefficientList[Exponent[%29, ℯ] /. (s + a) → s', s'], {r^2 + t^2 == 1, t^2 v1 + r^2 v2 == v', Γ == Γ'/t, μ == μ'/t^2}]
```

Out[ ]=
$$\left\{-\frac{q1^2}{2\,v'},\ \frac{q1\,\Gamma'}{v'},\ -\frac{\Gamma^2\,(v'+\mu')}{2\,\mu\,v'}\right\}$$

```
(* For Etilde convG term *)
ClearAll[a, b, c];

{cg, bg, ag} = CoefficientList[-(q2 r + p1 t)^2/(2 v1) - (-p1 r + q2 t)^2/(2 v2) - s^2 Γ^2/(2 v1) + 2 ⅈ π s (a + ⅈ (q2 r + p1 t) Γ/(2 π v1)), {q2}];

Sqrt[π]/(2 π Sqrt[v1] Sqrt[v2]) ℯ^(bg^2/(4 (-ag)) + cg)/Sqrt[-ag] (* Gaussian integral over position q2 *)
```

Out[ ]=
$$\frac{e^{2\,i\,a\,\pi\,s-\frac{p1^2\,t^2}{2\,v1}-\frac{p1^2\,r^2}{2\,v2}-\frac{p1\,s\,t\,\Gamma}{v1}-\frac{s^2\,\Gamma^2}{2\,v1}+\frac{\left(-\frac{p1\,r\,t}{v1}+\frac{p1\,r\,t}{v2}-\frac{r\,s\,\Gamma}{v1}\right)^2}{4\left(\frac{r^2}{2\,v1}+\frac{t^2}{2\,v2}\right)}}}{2\,\sqrt{\pi}\,\sqrt{v1}\,\sqrt{\frac{r^2}{2\,v1}+\frac{t^2}{2\,v2}}\,\sqrt{v2}}$$

In[ ]:=
```
Simplify[CoefficientList[Exponent[%70, ℯ], s], {r^2 + t^2 == 1, t^2 v1 + r^2 v2 == v', Γ == Γ'/t, μ == μ'/t^2}]
```

Out[ ]=
$$\left\{-\frac{p1^2}{2\,v'},\ 2\,i\,a\,\pi-\frac{p1\,\Gamma'}{v'},\ -\frac{(\Gamma')^2}{2\,v'}\right\}$$

The result is $\mu \to \mu t^2$, $\Gamma \to \Gamma t$, a→a, and v1 → t^2 v1 + r^2 v2 , when mode 2 is traced out

[Rslt] Vacuum loss

In[ ]:=
```
ProductTermLoss[t_, q_, p_, d_, i_, j_, v1_, Γ_, parity_] :=
  Chop[EconvG[Λ[v1, v1]/(4 v1) t^2, Γ t, (i + j)/(2 d) + parity/2, t^2 v1 + (1 - t^2) 1/2, q]] × Chop[EtconvG[Λ[v1, v1]/(4 v1) t^2, π Λ[v1, v1]/Γ t, (i - j)/(2 d) + parity/2, t^2 v1 + (1 - t^2) 1/2, p]];

WignerLoss[t_, q_, p_, d_, i_, j_, v1_, Γ_] := 1/Sqrt[norm[v1, v1, Γ, i, d] × norm[v1, v1, Γ, j, d]] Chop[Sum[ProductTermLoss[t, q, p, d, i, j, v1, Γ, parity], {parity, 0, 1}]];

ClearAll[mat, limits, range, v1, t, r, d, Γ];
limits = 2;
range = 60;
v1 = 0.05;
d = 2;
Γ = αd[d] d Sqrt[Λ[v1, v1]];
n = 4;
t = Sqrt[0.98];
r = Sqrt[1 - t^2];

(* List Density Plot - faster than Density plot *)
Quiet[
  mat = Table[Re[Chop[WignerLoss[t, Sqrt[π] ((x - 1) (limits/range) - limits), Sqrt[π] ((p - 1) (limits/range) - limits), d, 0, 0, v1, Γ]]], {p, 1, 2 range + 1}, {x, 1, 2 range + 1}];
 ];

mat = mat/(Abs[Total[mat, 2]] (limits/range)^2);

plot = ListDensityPlot[mat, DataRange → {{-limits, limits}, {-limits, limits}}, ColorFunction → (Blend[{Blue, White, Red}, (#/Max[{Abs[Max[mat]], Abs[Min[mat]]}] + 1)/2] &),
  ColorFunctionScaling → False, PlotRange → All, LabelStyle → {FontSize → Medium, FontFamily → "Arial", Black}, AxesStyle → {Thick, Black}, ImageSize → Medium,
  FrameLabel → {"q/√π", "p/√π"}]
```

Out[ ]=